

3. Demonstrations of Using *Maple* in Calculus and Differential Equations

In this second introductory section we will give demonstrations of how Maple can be used in calculus and differential equations. Later, as you work through some of the lab sections, it may be helpful to return to this section to see how some of the code in Maple is actually used.

We will begin with some calculus operations that include algebraic manipulations.

▼ 3.1 Calculus Examples

The techniques introduced above will now be applied to the fundamental concepts: limit, derivative, and integral, and their applications.

▼ Limits

> *restart*

The *Maple* command to compute $\lim_{x \rightarrow a} f(x)$ is `limit(f(x), x = a)`. A limit template can be entered in 2D input by typing *limit* then **[esc]/[enter]**. Press the **[tab]** key to move from one position to the next.

$$> \lim_{x \rightarrow 0} \frac{\sin(x)}{x} \quad 1 \quad (1.1.1)$$

$$> \lim_{x \rightarrow 0} (1 + x)^{\frac{1}{x}} \quad e \quad (1.1.2)$$

The infinity symbol is entered by typing *infinity* **[esc]**.

$$> \lim_{n \rightarrow \infty} \left(1 + \frac{x}{n}\right)^n \quad e^x \quad (1.1.3)$$

Right- and left-hand limits are returned when a "plus" or a "minus" is entered in the exponent of the limiting point.

$$> \lim_{x \rightarrow 0} \frac{\ln(x)}{x} \quad \text{undefined} \quad (1.1.4)$$

$$> \lim_{x \rightarrow 0^+} \frac{\ln(x)}{x} \quad -\infty \quad (1.1.5)$$

$$> \lim_{x \rightarrow 0^-} \frac{|x|}{x} \quad -1 \quad (1.1.6)$$

Using 1D input, the last limit is calculated like this.

$$> \text{limit}(\text{abs}(x)/x, x=0, \text{left});$$

$$-1 \quad (1.1.7)$$

Limits at infinity can also be handled.

$$> \lim_{x \rightarrow +\infty} \arctan(x)$$

$$\frac{1}{2} \pi \quad (1.1.8)$$

Here is the 1D input for the arctan limit.

$$> \text{limit}(\arctan(x), x=+\infty);$$

$$\frac{1}{2} \pi \quad (1.1.9)$$

>

▼ Difference Quotients and Derivatives

> restart

Given a function f , the slope of the secant line through $(x, f(x))$ and $(x+h, f(x+h))$ is called the difference quotient. As h approaches 0, these quantities converge to the slope of the tangent line at $(x, f(x))$. The slope of the secant line can be obtained with the Maple command

$$> m[\text{secant}] := \frac{f(x+h) - f(x)}{h}$$

$$m_{\text{secant}} := \frac{f(x+h) - f(x)}{h} \quad (1.2.1)$$

The difference quotient formula is now stored in table named m .

In practise, it is often necessary to use the *simplify* command to force *Maple* to simplify the difference quotient. For example,

$$> f(x) := x^3 - 3x^2 - 4$$

$$f := x \rightarrow x^3 - 3x^2 - 4 \quad (1.2.2)$$

Notice that in the previous command, *Maple* complains about parts of the expression being ambiguous, in this case, $f(x)$. If you intended this to define a function, select the function definition from the drop-down box that shows up. You should NOT really use $f(x)$ to define a function!

$$> m[\text{secant}]$$

$$\frac{(x+h)^3 - 3(x+h)^2 - x^3 + 3x^2}{h} \quad (1.2.3)$$

$$> \text{simplify}(m[\text{secant}])$$

$$3x^2 + 3xh + h^2 - 6x - 3h \quad (1.2.4)$$

$$> m[\text{tangent}] := \lim_{h \rightarrow 0} m[\text{secant}]$$


$$m_{\text{tangent}} := 3x^2 - 6x \quad (1.2.5)$$

The *Tangent* command from the *Student[Calculus I]* package provides a simple means to display a function and the tangent line at a point.

> with(*Student[Calculus I]*):

The optional equation *title* = "" is to suppress the (rather long) default title to the plot.

> `Tangent (f (x), x = 1, -1 ..3, output = plot, title = "")`



The expression for the derivative of the function f is obtained using the usual prime notation (2D input only). Make a prime by typing an apostrophe.

> $f'(x)$

$$3x^2 - 6x \quad (1.2.6)$$

The derivative operator: D , can also be used. The output to $D(f)$ is the derivative function.

> $D(f)$

$$x \rightarrow 3x^2 - 6x \quad (1.2.7)$$

Higher order derivative formulas are obtained using more primes, or the standard $f^{(n)}$ for the n^{th} derivative.

> $f''(x)$

$$6x - 6 \quad (1.2.8)$$

> $f^{(2)}(x)$

$$6x - 6 \quad (1.2.9)$$

Enter $f^{(2)}(x)$ using 2D input by typing $f^{(2)}$ [right arrow] (x) .

The *diff* procedure can be used to differentiate an expression.

> $\text{diff}(f(x), x)$

$$3x^2 - 6x \quad (1.2.10)$$

Type *diff* [esc] to enter a derivative template. The correct template for an *ordinary* derivative is the third choice on the contextual menu for *diff*.

> $\frac{d}{dx}f(x)$

$$3x^2 - 6x \quad (1.2.11)$$

Higher-order derivatives of *expressions* can be obtained either like this (second derivative with respect to x)

> $\text{diff}(xe^{ax}, x, x)$

$$2ae^{ax} + xa^2e^{ax} \quad (1.2.12)$$

or like this

$$\begin{aligned}
 &> \frac{d^2}{dx^2} (x e^{ax}) \\
 &\qquad 2 a e^{ax} + x a^2 e^{ax}
 \end{aligned} \tag{1.2.13}$$

Using 1D input the second derivative calculation is accomplished with the following entry.

$$\begin{aligned}
 &> \text{diff}(x*\exp(a*x), x, x); \\
 &\qquad 2 a e^{ax} + x a^2 e^{ax}
 \end{aligned} \tag{1.2.14}$$

Obtain the fifth-order derivative like this

$$\begin{aligned}
 &> \text{diff}(x e^{ax}, x\$5) \\
 &\qquad 5 a^4 e^{ax} + x a^5 e^{ax}
 \end{aligned} \tag{1.2.15}$$

or, using `diff[esc]`, like this

$$\begin{aligned}
 &> \frac{d^5}{dx^5} (x e^{ax}) \\
 &\qquad 5 a^4 e^{ax} + x a^5 e^{ax}
 \end{aligned} \tag{1.2.16}$$

Implicit differentiation is carried out in the same manner, provided that all functional dependencies are explicitly given in the equation. Here is an example for the function $y(x)$ defined implicitly by the following equation.

$$\begin{aligned}
 &> \text{eqn} := (x - 1)^4 = x^2 - y^2 \\
 &\qquad \text{eqn} := (x - 1)^4 = x^2 - y^2
 \end{aligned} \tag{1.2.17}$$

Make the substitution $y = y(x)$ to indicate that the equation is determining y as a function of x .

$$\begin{aligned}
 &> \text{ImpEqn} := \text{subs}(y = y(x), \text{eqn}) \\
 &\qquad \text{ImpEqn} := (x - 1)^4 = x^2 - y(x)^2
 \end{aligned} \tag{1.2.18}$$

Differentiate the implicit equation with respect to x .

$$\begin{aligned}
 &> \frac{d}{dx} \text{ImpEqn} \\
 &\qquad 4 (x - 1)^3 = 2 x - 2 y(x) \left(\frac{d}{dx} y(x) \right)
 \end{aligned} \tag{1.2.19}$$

Use the `isolate` command to isolate the derivative.

$$\begin{aligned}
 &> \text{isolate} \left(\%, \frac{d}{dx} y(x) \right) \\
 &\qquad \frac{d}{dx} y(x) = \frac{1}{2} \frac{-4 (x - 1)^3 + 2 x}{y(x)}
 \end{aligned} \tag{1.2.20}$$

The second derivative can be found like this.

$$\begin{aligned}
 &> y'' = \frac{d}{dx} \text{rhs}(\%) \\
 &\qquad \frac{d^2}{dx^2} y(x) = \frac{1}{2} \frac{-12 (x - 1)^2 + 2}{y(x)} - \frac{1}{2} \frac{(-4 (x - 1)^3 + 2 x) \left(\frac{d}{dx} y(x) \right)}{y(x)^2}
 \end{aligned} \tag{1.2.21}$$

Prime notation in Maple (2D input)

One prime on a free variable is interpreted by *Maple* as the derivative with respect to x . Two primes is the second derivative.

> y', y''

$$\frac{d}{dx} y(x), \frac{d^2}{dx^2} y(x) \quad (1.2.22)$$

Differentiation with respect to the t variable is indicated with a dot over the variable (Newton's notation). The dot is entered as a period.

> \dot{y}, \ddot{y}

$$\frac{d}{dt} y(t), \frac{d^2}{dt^2} y(t) \quad (1.2.23)$$

Overdot templates will be found on the Accents palette or, using the keyboard, by typing y [Control]-['] (this is [Control]-[shift]-[']) and then pressing the period key. Right-arrow your way back to the baseline. As usual, use the Command key on a Macintosh.

Of course, with *Maple* at our disposal, it is just as easy to obtain explicit derivative formulas. We can solve the original equation for y .

> $\text{soln} := \text{solve}(\text{eqn}, y)$

$$\text{soln} := \sqrt{-1 - x^4 + 4x^3 - 5x^2 + 4x}, -\sqrt{-1 - x^4 + 4x^3 - 5x^2 + 4x} \quad (1.2.24)$$

Apply the *diff* procedure to the first solution.

> $\frac{d}{dx}(\text{soln}[1])$

$$\frac{1}{2} \frac{-4x^3 + 12x^2 - 10x + 4}{\sqrt{-1 - x^4 + 4x^3 - 5x^2 + 4x}} \quad (1.2.25)$$

And here is the explicit second derivative of the second function defined by *eqn*. The *simplify* procedure is applied to the output to get it into one term.

> $\frac{d^2}{dx^2}(\text{soln}[2]) : \text{simplify}(\%)$

$$-\frac{2x^6 - 12x^5 + 27x^4 - 36x^3 + 30x^2 + 1 - 12x}{(-1 - x^4 + 4x^3 - 5x^2 + 4x)^{3/2}} \quad (1.2.26)$$

>

▼ Applications of Derivatives

▼ Linearization of a Function

> *restart*

The linearization of a function $f(x)$ at $x=a$ is the function $L(x) = f(a) + f'(a) \cdot (x - a)$. To create this function in Maple we enter

> $L(x) := f(a) + f'(a) \cdot (x - a)$

$$L := x \rightarrow f(a) + D(f)(a)(x - a) \quad (1.3.1.1)$$

Now, if we want the linearization formula for

$$> f(x) := \sqrt{1+x}$$

$$f := x \rightarrow \sqrt{1+x} \quad (1.3.1.2)$$

at

$$> a := 0$$

$$a := 0 \quad (1.3.1.3)$$

we need only evaluate L .

$$> L(x)$$

$$1 + \frac{1}{2}x \quad (1.3.1.4)$$

If the function and/or the point are changed


$$> f(x) := x \sin(x) : a := \frac{5}{2} :$$

then L still provides the correct linearization.

$$> L(x);$$

$$\text{plot}([f(x), L(x)], x = 1..4, \text{color} = \text{black}, \text{linestyle} = [1, 3])$$

$$\frac{5}{2} \sin\left(\frac{5}{2}\right) + \left(\sin\left(\frac{5}{2}\right) + \frac{5}{2} \cos\left(\frac{5}{2}\right)\right) \left(x - \frac{5}{2}\right)$$




To determine the largest interval on which the linearization displayed above differs from the function by no more than a specified amount, i.e., to estimate the largest δ such that $|x - a| < \delta$ implies $|f(x) - L(x)| < \epsilon$, we first look at the graph.

$$> \epsilon := 0.01;$$

$$\text{plot}([|f(x) - L(x)|, \epsilon], x = 2.4..2.6, \text{color} = \text{black}, \text{linestyle} = [1, 3])$$

$$\epsilon := 0.01$$



Click the left mouse button as close as possible to the leftmost of the two intersections of the curves in the plot. The box at the left end of the Context Bar shows the approximate coordinates of this point; this gives the point (2.42, 0.01). Repeating this process for the rightmost intersection gives (2.58, 0.01). This gives the estimate $\delta = 0.08$.

More accurate estimates of the intersection points will give more accurate estimates of δ . In this

case, Maple can find better approximations.

```
> leftpoint := fsolve(|f(x) - L(x)| = ε, x = 2 .. 2.5);
   rightpoint := fsolve(|f(x) - L(x)| = ε, x = 2.5 .. 3)
                        leftpoint := 2.419697538
                        rightpoint := 2.580447088
```

(1.3.1.5)


This yields

```
> δ := min(2.5 - leftpoint, rightpoint - 2.5)
                        δ := 0.080302462
```

(1.3.1.6)

To conclude, check that the error never exceeds ϵ over this interval.

```
> plot([|f(x) - L(x)|, ε], x = 2.5 - δ .. 2.5 + δ, color = black)
```



```
>
```

▼ Newton's Method

```
> restart
```

Newton's Method can be implemented in *Maple* to find an approximation to the smallest solution to $f(x) = 0$ with

```
> f(x) := cos(5 x) - x
                        f := x → cos(5 x) - x
```

(1.3.2.1)

to five decimal places by defining the auxiliary function

```
> g := unapply(x - f(x)/f'(x), x)
                        g := x → x - (cos(5 x) - x) / (-5 sin(5 x) - 1)
```

(1.3.2.2)

and an initial guess (use a graph, if necessary, to obtain a starting value)

```
> x0 := -1.0
                        x0 := -1.0
```

(1.3.2.3)

The next approximation is

```
> x1 := g(x0)
                        x1 := -0.7784735010
```

(1.3.2.4)

Note that we are using a table named x to store the approximate solutions.

Continuing in the same manner we can use a simple **for..do** loop to calculate three more approximations.

```
> for n from 2 to 4 do x_n := g(x_{n-1}) end do
                        x2 := -0.7677474241
```

$$x_3 := -0.7674935683$$

$$x_4 := -0.7674934212 \quad (1.3.2.5)$$

Thus, $x = 0.767493$ appears to be a 6-digit approximation for the smallest root. If this process had continued too much longer, it would be efficient to use the following modification of the loop.

$$\begin{aligned} &> \text{for } n \text{ from } 2 \text{ while } |x_{n-1} - x_{n-2}| \geq 10^{-6} \text{ do } x_n := g(x_{n-1}); \text{end do} \\ &\quad x_2 := -0.7677474241 \\ &\quad x_3 := -0.7674935683 \\ &\quad x_4 := -0.7674934212 \end{aligned} \quad (1.3.2.6)$$

▼ Maxima-Minima

Consider the following sequence of operations relating to the function $x(t) = e^{(-\alpha t)} \cos(\omega t + \theta)$. What information is obtained in each step?

$$\begin{aligned} &> \text{restart;} \\ &> x := \exp(-\alpha \cdot t) \cdot \cos(\omega \cdot t + \theta); \\ &\quad \# \text{ Note that } x \text{ is not declared to be a function but is an expression. We will refer to it} \\ &\quad \text{as } x \text{ below instead of } x(t). \\ &\quad x := e^{-\alpha t} \cos(\omega t + \theta) \end{aligned} \quad (1.3.3.1)$$

$$\begin{aligned} &> dx := \text{diff}(x, t); \\ &\quad dx := -\alpha e^{-\alpha t} \cos(\omega t + \theta) - e^{-\alpha t} \sin(\omega t + \theta) \omega \end{aligned} \quad (1.3.3.2)$$

$$\begin{aligned} &> dx2 := \text{diff}(x, t^2); \\ &\quad dx2 := \alpha^2 e^{-\alpha t} \cos(\omega t + \theta) + 2 \alpha e^{-\alpha t} \sin(\omega t + \theta) \omega - e^{-\alpha t} \cos(\omega t + \theta) \omega^2 \end{aligned} \quad (1.3.3.3)$$

$$\begin{aligned} &> \text{factor}(dx); \\ &\quad -e^{-\alpha t} (\alpha \cos(\omega t + \theta) + \sin(\omega t + \theta) \omega) \end{aligned} \quad (1.3.3.4)$$

$$\begin{aligned} &> \text{CriticalPoint} := \text{solve}(dx = 0, t); \\ &\quad \text{CriticalPoint} := -\frac{\theta + \arctan\left(\frac{\alpha}{\omega}\right)}{\omega} \end{aligned} \quad (1.3.3.5)$$

$$\begin{aligned} &> \text{subs}(t=\text{CriticalPoint}, dx2); \\ &\quad \alpha^2 e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right) \right)}{\omega}} \cos\left(-\arctan\left(\frac{\alpha}{\omega}\right)\right) + 2 \alpha e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right) \right)}{\omega}} \sin\left(-\arctan\left(\frac{\alpha}{\omega}\right)\right) \omega - e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right) \right)}{\omega}} \cos\left(-\arctan\left(\frac{\alpha}{\omega}\right)\right) \omega^2 \end{aligned} \quad (1.3.3.6)$$

$$> dx2 := \text{simplify}((1.3.3.6), \text{trig}); \# \text{ We have used the equation labels on the right.}$$

$$dx2 := - \frac{\alpha^2 e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right)\right)}{\omega}}}{\sqrt{\frac{\alpha^2}{\omega^2} + 1}} - \frac{e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right)\right)}{\omega}}}{\omega^2 \sqrt{\frac{\alpha^2}{\omega^2} + 1}} \quad (1.3.3.7)$$

Since the second derivative at the critical point will be negative (for real values of omega, alpha, and theta), this critical point must be a (local) maximum. The value of the function at this maximum is

> $X[\text{max}] := \text{simplify}(\text{subs}(t = \text{CriticalPoint}, x), \text{trig});$

$$X_{\text{max}} := \frac{e^{\frac{\alpha \left(\theta + \arctan\left(\frac{\alpha}{\omega}\right)\right)}{\omega}}}{\sqrt{\frac{\alpha^2}{\omega^2} + 1}} \quad (1.3.3.8)$$

Note the effect of [max] on the symbol X . It becomes a subscript. It is important to realize that the symbol X (upper-case) is not the same as the symbol x (lower case). The disastrous effect can be observed by changing the symbol above into lower case letters and re-executing the command and the next two commands below. Indefinite and definite integration is obtained in a natural way:

> $\text{Int}(x, t);$

$$\int e^{-\alpha t} \cos(\omega t + \theta) dt \quad (1.3.3.9)$$

> $\text{simplify}(\text{value}((1.3.3.9)));$

$$- \frac{e^{-\alpha t} (\alpha \cos(\omega t + \theta) - \sin(\omega t + \theta) \omega)}{\alpha^2 + \omega^2} \quad (1.3.3.10)$$

Note that the constant of integration is not included in the result. One can use the symbol palette to input the commands:

> $\text{simplify}\left(\int x dt\right);$

$$- \frac{e^{-\alpha t} (\alpha \cos(\omega t + \theta) - \sin(\omega t + \theta) \omega)}{\alpha^2 + \omega^2} \quad (1.3.3.11)$$

>

▼ Definite, Indefinite, and Improper Integrals

> restart

The *int* procedure is used to compute integrals in Maple. The syntax of an indefinite integral is $\text{int}(f(x), x)$.

> $\text{int}(\ln(x), x)$

(1.4.1)

$$x \ln(x) - x \quad (1.4.1)$$

Note that *Maple* does not add an arbitrary constant.

A definite integral over the interval $[a, b]$ is evaluated with the entry $\text{int}(f(x), x = a..b)$.

> $\text{int}(\ln(x), x = 1..4)$

$$-3 + 8 \ln(2) \quad (1.4.2)$$

Both these integrals can also be entered from the Expression palette. Alternatively, type int [esc] to obtain a contextual menu with even more integral templates.

> $\int x \sin(x) \, dx$

$$\sin(x) - x \cos(x) \quad (1.4.3)$$

> $\int_0^6 x \sin(x) \, dx$

$$\sin(6) - 6 \cos(6) \quad (1.4.4)$$

If approximate (decimal) limits are used, *Maple* evaluates the integral numerically.

> $\int_0^{6.0} x \sin(x) \, dx$

$$-6.040437218 \quad (1.4.5)$$

>

We will use *Maple* to find the function that satisfies the initial value problem $y' = 5e^{-3x}$, $y(0) = -10$. This problem can actually be solved in two ways.

Method 1. First find an antiderivative, including an arbitrary constant.

> $\text{eq1} := y = \int 5e^{-3x} dx + C$

$$\text{eq1} := y = -\frac{5}{3}e^{-3x} + C \quad (1.4.6)$$

Apply the initial condition to determine the appropriate value of C.

> $\text{eq2} := \text{subs}(x = 0, y = -10, \text{eq1})$

$$\text{eq2} := -10 = -\frac{5}{3}e^0 + C \quad (1.4.7)$$

Solve for C.

> $\text{isolate}(\text{eq2}, C)$

$$C = -\frac{25}{3} \quad (1.4.8)$$

Substitute this value into eq1 .

> $\text{subs}((1.4.8), \text{eq1})$

$$y = -\frac{5}{3}e^{-3x} - \frac{25}{3} \quad (1.4.9)$$

>

Method 2. Use a definite integral with lower limit 0 and upper limit x as the antiderivative. Add the initial value of y as the constant of integration.

Note the use of the dummy integration variable ξ instead of x . (ξ is Greek for x . Enter it by typing

xi [esc].) Maple will allow the use of the letter x as the integration variable, but this is not recommended.

$$> y = \int_0^x 5 e^{-3\xi} d\xi - 10$$

$$y = -\frac{5}{3} e^{-3x} - \frac{25}{3} \quad (1.4.10)$$

The `Student[Calculus I]` package contains a number of commands for the visualization and estimation of Riemann sums.

> with(Student[Calculus I])
 [AntiderivativePlot, AntiderivativeTutor, ApproximateInt, ApproximateIntTutor, ArcLength, ArcLengthTutor, Asymptotes, Clear, CriticalPoints, CurveAnalysisTutor, DerivativePlot, DerivativeTutor, DiffTutor, ExtremePoints, FunctionAverage, FunctionAverageTutor, FunctionChart, FunctionPlot, GetMessage, GetNumProblems, GetProblem, Hint, InflectionPoints, IntTutor, Integrand, InversePlot, InverseTutor, LimitTutor, MeanValueTheorem, MeanValueTheoremTutor, NewtonQuotient, NewtonsMethod, NewtonsMethodTutor, PointInterpolation, RiemannSum, RollesTheorem, Roots, Rule, Show, ShowIncomplete, ShowSolution, ShowSteps, Summand, SurfaceOfRevolution, SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem] (1.4.11)


To illustrate, consider the problem of approximating the value of $\int_1^2 x^3 + 1 dx$. Let

$$> f(x) := x^3 + 1$$

$$f := x \rightarrow x^3 + 1 \quad (1.4.12)$$

The midpoint approximation to this integral with 6 subdivisions can be visualized with

> ApproximateInt(f(x), x = 1..2, output = plot, method = midpoint, partition = 6, title = "")



The numerical approximation to the integral, the area of the six rectangles, is

> ApproximateInt(f(x), x = 1..2, output = sum, method = midpoint, partition = 6)

$$\frac{1}{6} \sum_{i=0}^5 \left(\left(\frac{13}{12} + \frac{1}{6} i \right)^3 + 1 \right) \quad (1.4.13)$$

> evalf (%)

$$4.739583334 \quad (1.4.14)$$

For right-hand sums use *method = right* and for left-hand sums, *method = left*. For the Trapezoidal and Simpson's Rules, use *trapezoid* and *simpson* respectively. For example,

$$> \text{evalf}(\text{ApproximateInt}(f(x), x = 1..2, \text{output} = \text{sum}, \text{method} = \text{trapezoid}, \text{partition} = 6)) \quad (1.4.15)$$

$$4.770833333$$

The exact value of the definite integral is

$$> \int_1^2 f(x) \, dx; \text{evalf}(\%)$$

$$\frac{19}{4} \quad (1.4.16)$$

$$4.750000000$$

While *Maple* is designed to be able to automatically evaluate many integrals, it can also be used to assist with techniques such as substitution and integration by parts. To prevent *Maple* from evaluating an integral, definite or indefinite, use *Int*, the inert form of *int*. The *value* command is used to force the symbolic evaluation of an inert command. A substitution is carried out with the *dchange* command in the *PDETools* package. The next entry defines an inert integral and names it *R*.

$$> R := \int_0^3 x \sqrt{3x+1} \, dx$$

$$R := \int_0^3 x \sqrt{3x+1} \, dx \quad (1.4.17)$$

Load the *PDETools* package (suppress output).

> with(PDETools) :

Define a change-of-variable equation and use *isolate* to generate the inverse change-of-variable equation.

> cv := u = 3x + 1;
icv := isolate(cv, x)

$$cv := u = 3x + 1$$

$$icv := x = \frac{1}{3} u - \frac{1}{3} \quad (1.4.18)$$

Apply *dchange* to the sequence *icv, R*.

> dchange(icv, R)

$$\int_1^{10} \frac{1}{3} \left(\frac{1}{3} u - \frac{1}{3} \right) \sqrt{u} \, du \quad (1.4.19)$$

> simplify(%)

$$\frac{1}{9} \int_1^{10} (u-1) \sqrt{u} \, du \quad (1.4.20)$$

> value(%)

$$\frac{100}{27} \sqrt{10} + \frac{4}{135} \quad (1.4.21)$$

To check the answer, compare it to the *Maple*-generated value of the original integral.

> value(R)


$$\frac{100}{27} \sqrt{10} + \frac{4}{135} \quad (1.4.22)$$

Integration-by-parts can be applied using the *IntTutor* in the *Student[Calculus I]* package. To apply it to the integral $\int x \sin(ax) \, dx$, evaluate the entry

> IntTutor(x sin(a x), x)

$$\int x \sin(ax) \, dx \quad (1.4.23)$$

Doing so produces the following dialogue allowing for the application of any one of a number of integration techniques.



Here's an example taken from the CRC Tables of Integrals.

> Int(u / sin(u), u = 0 ..Pi/2) :

> % = value(%);

$$\int_0^{\frac{1}{2}\pi} \frac{u}{\sin(u)} \, du = 2 \text{ Catalan} \quad (1.4.24)$$

> evalf(rhs(%));

1.831931188

(1.4.25)

>

▼ Applications of Integrals

▼ Arclength of a Smooth Curve

> restart

The definite integral for the arclength of a smooth curve is usually easy to setup but difficult to evaluate.

> $f := \frac{4\sqrt{2}}{3} \cdot x^{\frac{3}{2}} - 1$

$f := \frac{4}{3} \sqrt{2} x^{3/2} - 1$ (1.5.1.1)

> $\int_0^1 \sqrt{1 + \left(\frac{d}{dx} f\right)^2} dx$

$\frac{13}{6}$ (1.5.1.2)

Or, if the curve is entered in *Maple* as a function

> $f(x) := \frac{4\sqrt{2}}{3} \cdot x^{\frac{3}{2}} - 1$

$f := x \rightarrow \frac{4}{3} \sqrt{2} x^{3/2} - 1$ (1.5.1.3)

> $\int_0^1 \sqrt{1 + f(x)^2} dx$

$\frac{13}{6}$ (1.5.1.4)

Maple is able to evaluate many more integrals than most humans, but there are many arclength integrals that cannot be evaluated explicitly. If this occurs, or if *Maple*'s answer involves functions that are not familiar to you, use *evalf* to force a numerical approximation of the integral.

> $f(x) := \sin(x)$

$f := x \rightarrow \sin(x)$ (1.5.1.5)

> $\int_0^1 \sqrt{1 + f(x)^2} dx$

$\sqrt{2} \text{EllipticE}\left(\sqrt{1 - \cos(1)^2}, \frac{1}{2} \sqrt{2}\right)$ (1.5.1.6)

> evalf(%)

1.311442498

(1.5.1.7)

>

▼ *Improper Integrals*

> restart

Improper integrals are no problem for *Maple*. Here are three examples.

$$> \int_1^{\infty} x e^{-x} dx$$

$$2 e^{-1} \quad (1.5.2.1)$$

$$> \int_0^2 \frac{1}{\sqrt{4-x^2}} dx$$

$$\frac{1}{2} \pi \quad (1.5.2.2)$$

$$> \int_1^{\infty} \frac{1}{x} dx$$

$$\infty \quad (1.5.2.3)$$

$$> \text{int}(\exp(-u^2), u = 0 .. \text{infinity});$$

$$\frac{1}{2} \sqrt{\pi} \quad (1.5.2.4)$$

$$> \text{int}(\exp(-a \cdot u^2), u = 0 .. \text{infinity});$$

$$\lim_{u \rightarrow \infty} \frac{1}{2} \frac{\sqrt{\pi} \operatorname{erf}(\sqrt{a} u)}{\sqrt{a}} \quad (1.5.2.5)$$

Note that Maple is unable to evaluate this limit until something is known about the parameter a .

> about(a);

a:
nothing known about this object

> assume(a > 0);

> about(a);

Originally a, renamed a~:
is assumed to be: RealRange(Open(0),infinity)

$$> \text{int}(\exp(-a \cdot u^2), u = 0 .. \text{infinity});$$

$$\frac{1}{2} \frac{\sqrt{\pi}}{\sqrt{a\sim}} \quad (1.5.2.6)$$

We can remove the assumption on a by assigning the symbol to itself.

> a := 'a';

$$a := a \quad (1.5.2.7)$$

> about(a);

a:
nothing known about this object

>

▼ *Integration of a Rational Function*

Consider the proper rational function:

$$\begin{aligned}
 > f := \frac{x^2 - 26 \cdot x - 47}{x^5 + 5 \cdot x^3 + 3 \cdot x^4 + 11 \cdot x^2 - 20}; \\
 &f := \frac{x^2 - 26x - 47}{x^5 + 5x^3 + 3x^4 + 11x^2 - 20} \quad (1.5.3.1)
 \end{aligned}$$

The antiderivative of f can be obtained by a variety of different means. Let's examine a few and compare the information obtained in each case.

$$\begin{aligned}
 > F := \text{Int}(f, x); \\
 &F := \int \frac{x^2 - 26x - 47}{x^5 + 5x^3 + 3x^4 + 11x^2 - 20} dx \quad (1.5.3.2)
 \end{aligned}$$

The simplest, but least instructive, approach is to simply let *Maple* do all the work. Note that the constant of integration is not included in indefinite integrals.

$$\begin{aligned}
 > F = \text{value}(F); \\
 &\int \frac{x^2 - 26x - 47}{x^5 + 5x^3 + 3x^4 + 11x^2 - 20} dx = \frac{23}{27} \ln(x+2) - \frac{4}{3} \ln(x-1) \\
 &\quad + \frac{13}{54} \ln(x^2+5) + \frac{91}{135} \sqrt{5} \arctan\left(\frac{1}{5}x\sqrt{5}\right) + \frac{1}{3(x+2)} \quad (1.5.3.3)
 \end{aligned}$$

or alternatively,

$$\begin{aligned}
 > \text{int}(f, x); \\
 &\frac{23}{27} \ln(x+2) - \frac{4}{3} \ln(x-1) + \frac{13}{54} \ln(x^2+5) + \frac{91}{135} \sqrt{5} \arctan\left(\frac{1}{5}x\sqrt{5}\right) \\
 &\quad + \frac{1}{3(x+2)} \quad (1.5.3.4)
 \end{aligned}$$

To emphasize integration techniques while still avoiding algebraic complications, we can utilize *Maple* to determine the partial fraction expansion of the integrand.

$$\begin{aligned}
 > fpf := \text{convert}(f, \text{parfrac}, x); \\
 &fpf := \frac{1}{27} \frac{13x+91}{x^2+5} - \frac{1}{3(x+2)^2} - \frac{4}{3(x-1)} + \frac{23}{27(x+2)} \quad (1.5.3.5)
 \end{aligned}$$

From this partial fraction decomposition it is now a relatively simple matter to compute the integral BY HAND. The result can be compared with the previous answer. But let *Maple* do it.

$$\begin{aligned}
 > \text{int}(fpf, x); \\
 &\frac{23}{27} \ln(x+2) - \frac{4}{3} \ln(x-1) + \frac{13}{54} \ln(x^2+5) + \frac{91}{135} \sqrt{5} \arctan\left(\frac{1}{5}x\sqrt{5}\right) \quad (1.5.3.6)
 \end{aligned}$$

$$+ \frac{1}{3(x+2)}$$

The third and final approach to this problem will be the most detailed. The purpose here is to emphasize the process of finding a partial fraction expansion without fear of unmanageable algebra. The key here is to enter the correct form for the partial fraction decomposition. This form depends upon the irreducible factors in the denominator of f .

> `factor(denom(f));`

$$(x-1)(x^2+5)(x+2)^2 \quad (1.5.3.7)$$

The appropriate general form for the decomposition will be

$$\begin{aligned} > \text{FORM} := \frac{a}{x-1} + \frac{b}{x+2} + \frac{c}{(x+2)^2} + \frac{d+e \cdot x}{x^2+5}; \\ \text{FORM} &:= \frac{a}{x-1} + \frac{b}{x+2} + \frac{c}{(x+2)^2} + \frac{d+e x}{x^2+5} \end{aligned} \quad (1.5.3.8)$$

The second step is to determine the (linear) equations that define the constants a , b , c , d , and e .

> `eqn := f = FORM;`

$$\begin{aligned} \text{eqn} &:= \frac{x^2 - 26x - 47}{x^5 + 5x^3 + 3x^4 + 11x^2 - 20} = \frac{a}{x-1} + \frac{b}{x+2} + \frac{c}{(x+2)^2} \\ &+ \frac{d+e x}{x^2+5} \end{aligned} \quad (1.5.3.9)$$

> `simplify(eqn * denom(f));`

$$\begin{aligned} x^2 - 26x - 47 &= a x^4 + 4 a x^3 + 9 a x^2 + 20 a x + 20 a + b x^4 + b x^3 + 3 b x^2 \\ &+ 5 b x - 10 b + c x^3 - c x^2 + 5 c x - 5 c + d x^3 + 3 d x^2 - 4 d + e x^4 \\ &+ 3 e x^3 - 4 e x \end{aligned} \quad (1.5.3.10)$$

This equation is true for all x . We can collect common powers of x to identify the coefficients.

> `collect((1.5.3.10), x);`

$$\begin{aligned} x^2 - 26x - 47 &= (a + e + b) x^4 + (4 a + 3 e + c + d + b) x^3 + (9 a + 3 b - c \\ &+ 3 d) x^2 + (-4 e + 5 b + 5 c + 20 a) x - 5 c - 10 b - 4 d + 20 a \end{aligned} \quad (1.5.3.11)$$

> `eqns := [coeffs(rhs((1.5.3.11)), x)];`

$$\begin{aligned} \text{eqns} &:= [-5 c - 10 b - 4 d + 20 a, -4 e + 5 b + 5 c + 20 a, 9 a + 3 b - c \\ &+ 3 d, 4 a + 3 e + c + d + b, a + e + b] \end{aligned} \quad (1.5.3.12)$$

Notice that the coefficients are not necessarily ordered in increasing powers of x . We obtain the following system of 5 linear equations in 5 unknowns.

> `systemofequations := {eqns[1] = -47, eqns[2] = -26, eqns[3] = 1, eqns[4] = 0, eqns[5] = 0};`

$$\text{systemofequations} := \{a + e + b = 0, 9 a + 3 b - c + 3 d = 1, -5 c - 10 b - 4 d \quad (1.5.3.13)$$

$$+ 20 a = -47, -4 e + 5 b + 5 c + 20 a = -26, 4 a + 3 e + c + d + b = 0\}$$

The solution to these equations is

$$\begin{aligned} &> \text{COEFFS} := \text{solve}(\text{systemofequations}, \{a, b, c, d, e\}); \\ \text{COEFFS} &:= \left\{ a = -\frac{4}{3}, b = \frac{23}{27}, c = -\frac{1}{3}, d = \frac{91}{27}, e = \frac{13}{27} \right\} \end{aligned} \quad (1.5.3.14)$$

Substituting this result, which is a set, into the original form for the partial fraction decomposition yields

$$\begin{aligned} &> \text{subs}(\text{COEFFS}, \text{FORM}); \\ &-\frac{4}{3(x-1)} + \frac{23}{27(x+2)} - \frac{1}{3(x+2)^2} + \frac{\frac{13}{27}x + \frac{91}{27}}{x^2 + 5} \end{aligned} \quad (1.5.3.15)$$

This agrees with (1.5.3.5). The integral is now much simpler to evaluate by hand, but here it is one last time according to *Maple*.

$$\begin{aligned} &> \text{int}((1.5.3.15), x); \\ &\frac{23}{27} \ln(x+2) - \frac{4}{3} \ln(x-1) + \frac{13}{54} \ln(x^2+5) + \frac{91}{135} \sqrt{5} \arctan\left(\frac{1}{5}x\sqrt{5}\right) \\ &+ \frac{1}{3(x+2)} \end{aligned} \quad (1.5.3.16)$$

Note that this same outline can be used on ANY partial fraction problem. The key steps are

- knowing the correct form for the partial fraction decomposition and
- understanding how to identify the conditions that the coefficients must satisfy.

These are (relatively) high level concepts; the user is relieved of the rather complicated mechanical manipulations.

>

▼ Multiple Integrals

> restart

The *int* and *Int* commands can be used to construct iterated integrals corresponding to double and triple integrals.

> Int(Int(exp(x-y), x=y..1), y=0..1);
value(%)

$$\int_0^1 \int_y^1 e^{x-y} dx dy = e - 2 \quad (1.5.4.1)$$

Using single integral templates, one inside the other, this integral calculation looks like this (2D input only).

$$> \int_0^1 \int_y^1 e^{x-y} dx dy$$

e- 2

(1.5.4.2)

The area inside of the ellipse $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$ can be found by parametrizing the ellipse in polar coordinates: $x = a r \cos(\theta)$, $y = b r \sin(\theta)$ with $0 \leq r \leq 1$ and $0 \leq \theta \leq 2\pi$. The Jacobian of this change-of-variables is $a \cdot b \cdot r$, so the area of the ellipse is

$$> \int_0^{2\pi} \int_0^1 a \cdot b \cdot r \, dr \, d\theta$$

a b π

(1.5.4.3)

Note that this result reduces to the familiar area of a circle when $a = b$.

>

▼ 3.2 Differential Equations Demonstrations

In this section we will demonstrate the use of *Maple* in working with differential equations. The following topics are covered in the following subsections.

- Direction Fields and Graphical Solutions (Sections 1.3 and 5.4 of Nagle/Saff/Snider)
- Symbolic Solutions to Second Order Ordinary Differential Equations and Initial-Value Problems (Chapter 4 of Nagle/Saff/Snider)
- Expressions and Functions (Chapter 4 of Nagle/Saff/Snider)
- Systems of Ordinary Differential Equations (Sections 5.2, 5.4 and 5.5 of Nagle/Saff/Snider)
- Numeric Solutions (Sections 3.6 and 5.3 of Nagle/Saff/Snider)
- Series Solutions (Chapter 8 of Nagle/Saff/Snider)
- Laplace Transforms (Chapter 7 of Nagle/Saff/Snider)

It is a good habit to remember to use *restart* when starting a new problem.

> *restart*;

Another good habit is to load the [plots](#) and [DEtools](#) packages at the top of all worksheets, just in case they will be needed at some point in the analysis.

> *with*(*plots*);

[*animate*, *animate3d*, *animatecurve*, *arrow*, *changecoords*, *complexplot*, *complexplot3d*, (2.1)

conformal, *conformal3d*, *contourplot*, *contourplot3d*, *coordplot*, *coordplot3d*, *densityplot*, *display*, *dualaxisplot*, *fieldplot*, *fieldplot3d*, *gradplot*, *gradplot3d*, *implicitplot*, *implicitplot3d*, *inequal*, *interactive*, *interactiveparams*, *intersectplot*, *listcontplot*, *listcontplot3d*, *listdensityplot*, *listplot*, *listplot3d*, *loglogplot*, *logplot*, *matrixplot*, *multiple*, *odeplot*, *pareto*, *plotcompare*, *pointplot*, *pointplot3d*, *polarplot*, *polygonplot*, *polygonplot3d*, *polyhedra_supported*, *polyhedraplot*, *rootlocus*, *semilogplot*, *setcolors*, *setoptions*, *setoptions3d*, *spacecurve*, *sparsematrixplot*, *surfdata*, *textplot*, *textplot3d*, *tubeplot*]

> *with*(*DEtools*);

(2.2)

(2.2)

[*AreSimilar*, *DEnormal*, *DEplot*, *DEplot3d*, *DEplot_polygon*, *DFactor*, *DFactorLCLM*, *DFactorsols*, *Dchangevar*, *FunctionDecomposition*, *GCRD*, *Gosper*, *Heunsols*, *Homomorphisms*, *IVPsol*, *IsHyperexponential*, *LCLM*, *MeijerGsols*, *MultiplicativeDecomposition*, *ODEInvariants*, *PDEchangecoords*, *PolynomialNormalForm*, *RationalCanonicalForm*, *ReduceHyperexp*, *RiemannPsols*, *Xchange*, *Xcommutator*, *Xgauge*, *Zeilberger*, *abelsol*, *adjoint*, *autonomous*, *bernoullisol*, *buildsol*, *buildsym*, *canoni*, *caseplot*, *casesplit*, *checkrank*, *chinishol*, *clairautsol*, *constcoeffsols*, *convertAlg*, *convertsys*, *dalembertsol*, *dcoeffs*, *de2diffop*, *dfieldplot*, *diff_table*, *diffop2de*, *dperiodic_sols*, *dpolyform*, *dsubs*, *eigenring*, *endomorphism_charpoly*, *equinv*, *eta_k*, *eulersols*, *exactsol*, *expsols*, *exterior_power*, *firint*, *firtest*, *formal_sol*, *gen_exp*, *generate_ic*, *genhomosol*, *gensys*, *hamilton_eqs*, *hypergeomsols*, *hyperode*, *indiciaeq*, *infsen*, *initialdata*, *integrate_sols*, *intfactor*, *invariants*, *kovacicols*, *leftdivision*, *liesol*, *line_int*, *linearsol*, *matrixDE*, *matrix_riccati*, *maxdimsystems*, *moser_reduce*, *muchange*, *mult*, *mutest*, *newton_polygon*, *normalG2*, *ode_int_y*, *ode_y1*, *odeadvisor*, *odepde*, *parametricsol*, *particularsol*, *phaseportrait*, *poincare*, *polysols*, *power_equivalent*, *rational_equivalent*, *ratsols*, *redode*, *reduceOrder*, *reduce_order*, *regular_parts*, *regularsp*, *remove_RootOf*, *riccati_system*, *riccatisol*, *rifread*, *rifsimp*, *rightdivision*, *rtaylor*, *separablesol*, *singularities*, *solve_group*, *super_reduce*, *symgen*, *symmetric_power*, *symmetric_product*, *symtest*, *transinv*, *translate*, *untranslate*, *varparam*, *zoom*]

▼ Direction Fields and Graphical Solutions (Sections 1.3 and 5.4 of Nagle/Saff/Snider)

Many introductory courses begin by trying to develop the student's understanding of what a differential equation is, what it means for a function to solve an ODE, and how to perform some analysis directly from the differential equation. Graphical methods are commonly employed in these discussions. The *Maple* command [DEplot](#), from the *DEtools* package, provides a comprehensive interface for most graphical needs.

To begin, consider the first order (linear) differential equation


$$\begin{aligned} > \text{ODE} := \text{diff}(y(x), x) = x^2 - y(x); \\ &\text{ODE} := \frac{d}{dx} y(x) = x^2 - y(x) \end{aligned} \quad (2.1.1)$$

Note that this is the first example used in Section 1.3 of Nagle/Saff/Snider. The next few commands reproduce a few of the figures displayed on p.16-18.

The first three arguments to *DEplot* must provide, in order, a single n -th order ODE or a system of n first-order ODEs, a dependent variable or a list or set of dependent variables, and a range for the independent variable. The specific content of the plot is determined by all subsequent arguments to *DEplot*.


The most basic use of DEplot is to display a direction field for a differential equation. A request for a direction field is made by specifying both a range for the independent variable and the **arrows = option**.

```
> DEplot(ODE, y(x), x = -3 .. 3, y = -5 .. 5, arrows = thin, title
= Direction Field for y'=x^2-y);
```



The solution curves through specified points are produced when the fourth argument is a set of initial conditions. If the **arrows = none** is included instead of **arrows = thin**, only the solution curves are displayed. The **linecolor** gives the colors of the solution curves.

```
> DEplot(ODE, y(x), x = -1 .. 5, {[0, -3], [0, -2], [0, -1], [0, 0], [0, 1], [0, 2], [0, 3]},
arrows = thin, title = 'Solution Curves for y'=x^2-y', linecolor = [black, blue, black,
blue, black, blue, black]);
```




Only slight alterations are needed for higher-order equations, or for systems. To illustrate, consider the predator-prey system

$$\begin{aligned}
 &> \text{SYS} := [\text{diff}(P(t), t) = p(t) \cdot P(t) - P(t), \text{diff}(p(t), t) = 2 \cdot p(t) - 2 \cdot P(t) \\
 &\quad \cdot p(t)]; \\
 &\text{SYS} := \left[\frac{d}{dt} P(t) = p(t) P(t) - P(t), \frac{d}{dt} p(t) = 2 p(t) - 2 p(t) P(t) \right] \quad (2.1.2)
 \end{aligned}$$

where P denotes the size of the predator population, p is the size of the prey population and t is time.


A sampling of solution curves in phase-space can be created by specifying a set of initial conditions using the sequence operator [§](#) and the desired *scene*. (The subcommand $\{ [0, 1, k/2] \mid k = 2..6 \}$ below is a shortcut to generate the initial conditions $\{ [0, 1, 1], [0, 1, 3/2], [0, 1, 4/2], [0, 1, 5/2], [0, 1, 6/2], [0, 1, 3/2] \}$.) The *stepsize*=0.1 is included because the default stepsize produces a rather rough graph.

> `DEplot(SYS, [P(t), p(t)], 0..4, { [0, 1, k/2] $ k = 2..6 }, scene = [P, p], stepsize = 0.1, title = 'Predator-Prey: Phase Portrait', linecolor = BLACK);`



Plots of the individual solutions can be obtained by changing the scene. Here, individual plots of the predator and prey are created, and then displayed in a single plot using the [display](#) command from the plots package.

```
> plotp := DEplot( SYS, [ P(t), p(t) ], 0..10, { [0, 1, 2] }, scene = [t, p], stepsize = 0.1,
    linecolor = black) :
> plotP := DEplot( SYS, [ P(t), p(t) ], 0..10, { [0, 1, 2] }, scene = [t, P], stepsize = 0.1,
    linecolor = blue) :
> display( { plotp, plotP }, title = `Predator and Prey vs. Time`);
```



This is only the beginning of what can be done with the [DEplot](#) and [DEtools](#) package. Please consult the on-line help for additional information and a complete description of the available options and examples.

```
> ?DEtools
> ?DEtools[DEplot]
>
```

▼ Symbolic Solutions to Second Order ODEs and IVPs (*Chapter 4 of Nagle/Saff/Snider*)

While *Maple* is quite happy to accept the inputs to its commands in almost any form, it is practically, esthetically, and pedagogically preferable to use descriptive names for the individual parts of a problem (e.g., the differential equation and boundary condition). To illustrate, let's find the general solution to the non-homogeneous second order linear ODE $x'' + 4x' + 4x = 2t \cdot e^{-2t}$.

The differential equation can be entered as

$$\begin{aligned}
 &> ODE := \text{diff}(x(t), t\$2) + 4 \cdot \text{diff}(x(t), t) + 4 \cdot x(t) = 2 \cdot t \cdot \exp(-2 \cdot t); \\
 &ODE := \frac{d^2}{dt^2} x(t) + 4 \left(\frac{d}{dt} x(t) \right) + 4 x(t) = 2 t e^{-2 t}
 \end{aligned} \tag{2.2.1}$$

and the general solution is found using [dsolve](#).

$$\begin{aligned}
 &> GSOLN := \text{dsolve}(ODE, x(t)); \\
 &GSOLN := x(t) = e^{-2 t} _C2 + t e^{-2 t} _C1 + \frac{1}{3} t^3 e^{-2 t}
 \end{aligned} \tag{2.2.2}$$

Note that *Maple* introduces the symbols `_C1` and `_C2` as arbitrary constants. Any symbol beginning with an underscore is effectively reserved for use only by Maple library code. It is not available to users. Failure to observe this rule can lead to unexpected results.

The backslash key: `\`, is the escape key to enter certain symbols in 2D input. For example, because the underscore is used to enter a subscript, the entry `_C1` is made by typing `_C1`.

The solution to an initial value problem can be obtained by including initial conditions in a set containing the differential equation. For example, the initial conditions $x(1) = 0$, $x'(1) = 1$ can be implemented as

$$\begin{aligned}
 &> IC := x(1) = 0, D(x)(1) = 1; \\
 &IC := x(1) = 0, D(x)(1) = 1
 \end{aligned} \tag{2.2.3}$$

Note the use of [D](#) in the specification of the derivative condition (higher order derivative conditions can be specified using [@@](#), Maple's composition operator, e.g., $(D@@2(x))(0) = 3$). Then, the initial value problem is

$$\begin{aligned}
 &> IVP := \{ ODE, IC \}; \\
 &IVP := \left\{ \frac{d^2}{dt^2} x(t) + 4 \left(\frac{d}{dt} x(t) \right) + 4 x(t) = 2 t e^{-2 t}, x(1) = 0, D(x)(1) = 1 \right\}
 \end{aligned} \tag{2.2.4}$$

and its solution is found using [dsolve](#).

$$\begin{aligned}
 &> SOLN := \text{dsolve}(IVP, x(t)); \\
 &SOLN := x(t) = \frac{1}{3} \frac{e^{-2 t} (-3 + 2 e^{-2})}{e^{-2}} - \frac{t e^{-2 t} (e^{-2} - 1)}{e^{-2}} + \frac{1}{3} t^3 e^{-2 t}
 \end{aligned} \tag{2.2.5}$$

>

▼ Expressions and Functions (*Chapter 4 of Nagle/Saff/Snider*)

Notice how the output from **dsolve** is a *Maple* equation (the reason for this choice will be apparent when considering the solution to a system of differential equations). While there are specific cases where these equations are useful, most circumstances call for either an expression or a function. The key is to understand the structure of the output from **dsolve**.

Let's illustrate with the same example as above, with initial conditions specified, so that the constants refer to the values of the solution and its first derivative at $t = 1$.

$$\begin{aligned} > IC := \{ x(1) = c_1, D(x)(1) = c_2 \}; \\ &\quad IC := \{ x(1) = c_1, D(x)(1) = c_2 \} \end{aligned} \quad (2.3.1)$$

Note that these initial conditions are specified as a set, not an expression sequence. Thus, different manipulations are required to express the final IVP as a set.

$$\begin{aligned} > IVP := \{ ODE \} \text{ union } IC; \\ IVP := \left\{ \frac{d^2}{dt^2} x(t) + 4 \left(\frac{d}{dt} x(t) \right) + 4 x(t) = 2 t e^{-2t}, x(1) = c_1, D(x)(1) = c_2 \right\} \end{aligned} \quad (2.3.2)$$

It is easiest to obtain the solution as an expression; it's just the right-hand side of the object that *Maple* returns from [dsolve](#).

$$\begin{aligned} > X := rhs(dsolve(IVP, x(t))); \\ X := \frac{1}{3} \frac{e^{-2t} (-3c_2 - 3c_1 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t} (-c_2 - 2c_1 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.3)$$

When a Maple function is desired, it is most expedient to use [unapply](#).

$$\begin{aligned} > XX := unapply(rhs(dsolve(IVP, x(t))), t); \\ XX := t \rightarrow \frac{1}{3} \frac{e^{-2t} (-3c_2 - 3c_1 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t} (-c_2 - 2c_1 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.4)$$

To illustrate a potential advantage of the use of ***unapply***, consider a situation in which you wish to include the constants as arguments to the solution function, e.g., $x(t, c_1, c_2)$, where c_1 and c_2 are respectively, the values of the function and its first derivative at $t = 1$. Then you would use

$$\begin{aligned} > XXX := unapply(rhs(dsolve(IVP, x(t))), t, c_1, c_2); \\ XXX := (t, c_1, c_2) \rightarrow \frac{1}{3} \frac{e^{-2t} (-3c_2 - 3c_1 + 2e^{-2})}{e^{-2}} \\ &\quad - \frac{te^{-2t} (-c_2 - 2c_1 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.5)$$

and the resulting function, *XXX*, can be used to find the solution passing through the point (1, 0) with unit slope as follows

$$\begin{aligned} > XXX(t, 0, 1); \\ &\quad \frac{1}{3} \frac{e^{-2t} (-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t} (e^{-2} - 1)}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.6)$$

The same computation using *X* and *XX* would appear as

$$\begin{aligned} &> \text{subs}(c_1 = 0, c_2 = 1, X); \\ &\quad \frac{1}{3} \frac{e^{-2t}(-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(e^{-2} - 1)}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.7)$$

$$\begin{aligned} &> \text{subs}(c_1 = 0, c_2 = 1, XX(t)); \\ &\quad \frac{1}{3} \frac{e^{-2t}(-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(e^{-2} - 1)}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.8)$$

Next, consider the problem of verifying that the solution is correct.

$$\begin{aligned} &> \text{simplify}(\text{subs}(x(t) = X, ODE)); \\ &\quad 2te^{-2t} = 2te^{-2t} \end{aligned} \quad (2.3.9)$$

$$\begin{aligned} &> \text{simplify}(\text{subs}(x(t) = XX(t), ODE)); \\ &\quad 2te^{-2t} = 2te^{-2t} \end{aligned} \quad (2.3.10)$$

$$\begin{aligned} &> \text{simplify}(\text{subs}(x(t) = XXX(t, c_1, c_2), ODE)); \\ &\quad 2te^{-2t} = 2te^{-2t} \end{aligned} \quad (2.3.11)$$

To conclude this discussion, let's plot the solutions that have critical points (i.e., $c_2 = 0$) at (1,-1), (1,1), and (1,3) using each of X , XX and XXX .

$$\begin{aligned} &> S1 := \{ \text{seq}(\text{subs}(c_1 = C, c_2 = 0, X), C = \{-1, 1, 3\}) \}; \\ S1 &:= \left\{ \frac{1}{3} \frac{e^{-2t}(-9 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(-6 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t}, \right. \\ &\quad \frac{1}{3} \frac{e^{-2t}(-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(-2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t}, \\ &\quad \left. \frac{1}{3} \frac{e^{-2t}(3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \right\} \end{aligned} \quad (2.3.12)$$

$$\begin{aligned} &> S2 := \{ \text{seq}(\text{subs}(c_1 = C, c_2 = 0, XX('t')), C = \{-1, 1, 3\}) \}; \\ &\quad \#The use of 't' is explained below. \\ S2 &:= \left\{ \frac{1}{3} \frac{e^{-2t}(-9 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(-6 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t}, \right. \\ &\quad \frac{1}{3} \frac{e^{-2t}(-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(-2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t}, \\ &\quad \left. \frac{1}{3} \frac{e^{-2t}(3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \right\} \end{aligned} \quad (2.3.13)$$


$$\begin{aligned} &> S3 := \{ \text{seq}(XXX('t', C, 0), C = \{-1, 1, 3\}) \}; \\ S3 &:= \left\{ \frac{1}{3} \frac{e^{-2t}(-9 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t}(-6 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t}, \right. \end{aligned} \quad (2.3.14)$$

$$\frac{1}{3} \frac{e^{-2t} (-3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t} (-2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t},$$

$$\left. \frac{1}{3} \frac{e^{-2t} (3 + 2e^{-2})}{e^{-2}} - \frac{te^{-2t} (2 + e^{-2})}{e^{-2}} + \frac{1}{3} t^3 e^{-2t} \right\}$$

Note that the use of single quotes (') is needed so that the evaluation of t is delayed until after the **subs** command is completed.

> plot(S3, t = 0 .. 3, title = 'Function: XXX(t,c1,c2)');



The structure of this solution (linear combination of a basis of solutions to the homogeneous equation plus a particular solution) is evident in the solutions we have found. The individual components can be easily extracted and manipulated. Here we demonstrate how a general solution is used to determine the solution of a specific initial-value problem. Once the general solution is found,

> GSOLN := rhs(dsolve(ODE, x(t)));

$$GSOLN := e^{-2t} _C2 + te^{-2t} _C1 + \frac{1}{3} t^3 e^{-2t} \quad (2.3.15)$$

the two equations that must be satisfied by $_C1$ and $_C2$ can be constructed:

$$\begin{aligned} > eq1 := \text{subs}(t = 1, GSOLN = 3); eq2 := \text{subs}(t = 1, \text{diff}(GSOLN, t) = 0); \\ &eq1 := e^{-2} _C2 + e^{-2} _C1 + \frac{1}{3} e^{-2} = 3 \\ &eq2 := -2 e^{-2} _C2 - e^{-2} _C1 + \frac{1}{3} e^{-2} = 0 \end{aligned} \quad (2.3.16)$$

and solved

$$\begin{aligned} > solC := \text{solve}(\{eq1, eq2\}, \{_C1, _C2\}); \\ &solC := \left\{ _C1 = -\frac{-6 + e^{-2}}{e^{-2}}, _C2 = \frac{1}{3} \frac{-9 + 2 e^{-2}}{e^{-2}} \right\} \end{aligned} \quad (2.3.17)$$

These solutions obviously simplify (but require an application of `expand` to force all simplifications)

$$\begin{aligned} > solC := \text{simplify}(\text{expand}(solC)); \\ &solC := \left\{ _C1 = 6 e^2 - 1, _C2 = -3 e^2 + \frac{2}{3} \right\} \end{aligned} \quad (2.3.18)$$

Floating point approximations to these constants are obtained using [evalf](#):

$$\begin{aligned} > \text{evalf}(solC); \\ &\{ _C1 = 43.33433659, _C2 = -21.50050163 \} \end{aligned} \quad (2.3.19)$$

These values can be substituted back to $GSOLN$ but the exact values are preferred. The solution to the IVP is found to be

$$\begin{aligned} > IVPsoln := \text{factor}(\text{subs}(solC, GSOLN)); \\ &IVPsoln := \frac{1}{3} e^{-2t} (-9 e^2 + 2 + 18 t e^2 - 3 t + t^3) \end{aligned} \quad (2.3.20)$$

$$\begin{aligned} > \text{simplify}(IVPsoln - \text{subs}(c_1 = 3, c_2 = 0, X)); \\ &0 \end{aligned} \quad (2.3.21)$$

$$\begin{aligned} > \text{simplify}(IVPsoln - \text{subs}(c_1 = 3, c_2 = 0, XX(t))); \\ &0 \end{aligned} \quad (2.3.22)$$

$$\begin{aligned} > \text{simplify}(IVPsoln - XXX(t, 3, 0)); \\ &0 \end{aligned} \quad (2.3.23)$$

To conclude, let's see three different ways in which the terms contributing to the homogeneous and particular solutions can be obtained. Working directly from the general solution we find

$$\begin{aligned} > Xp := \text{subs}(_C1 = 0, _C2 = 0, GSOLN); \\ &Xp := \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.24)$$

$$\begin{aligned} > Xh := GSOLN - Xp; \\ &Xh := e^{-2t} _C2 + t e^{-2t} _C1 \end{aligned} \quad (2.3.25)$$

$$> Xl := \text{subs}(_C1 = 1, _C2 = 0, Xh);$$

$$X1 := t e^{-2t} \quad (2.3.26)$$

$$\begin{aligned} > X2 := \text{subs}(_C1 = 0, _C2 = 1, Xh); \\ X2 &:= e^{-2t} \end{aligned} \quad (2.3.27)$$

Alternatively, the same information can be obtained directly from *dsolve*.

$$\begin{aligned} > ODEh := \text{lhs}(ODE) = 0; \\ ODEh &:= \frac{d^2}{dt^2} x(t) + 4 \left(\frac{d}{dt} x(t) \right) + 4 x(t) = 0 \end{aligned} \quad (2.3.28)$$

$$\begin{aligned} > IC1 := x(0) = 0, D(x)(0) = 1; X1 := \text{rhs}(\text{dsolve}(\{ODEh, IC1\}, x(t))); \\ IC1 &:= x(0) = 0, D(x)(0) = 1 \\ X1 &:= t e^{-2t} \end{aligned} \quad (2.3.29)$$

$$\begin{aligned} > IC2 := x(0) = 1, D(x)(0) = 0; X2 := \text{rhs}(\text{dsolve}(\{ODEh, IC2\}, x(t))); \\ IC2 &:= x(0) = 1, D(x)(0) = 0 \\ X2 &:= e^{-2t} + 2 t e^{-2t} \end{aligned} \quad (2.3.30)$$

$$\begin{aligned} > ICp := x(0) = 0, D(x)(0) = 0; Xp := \text{rhs}(\text{dsolve}(\{ODE, ICp\}, x(t))); \\ ICp &:= x(0) = 0, D(x)(0) = 0 \\ Xp &:= \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.31)$$

Although, to be honest, in this case it is probably simplest to explicitly identify the appropriate terms in the general solution and explicitly define the components of the solution.

$$\begin{aligned} > x_1 &:= \exp(-2 \cdot t); x_2 := t * \exp(-2 \cdot t); x_p := \frac{t^3}{3} \cdot \exp(-2 \cdot t); \\ x_1 &:= e^{-2t} \\ x_2 &:= t e^{-2t} \\ x_p &:= \frac{1}{3} t^3 e^{-2t} \end{aligned} \quad (2.3.32)$$

This example hopefully illustrates that while almost all manipulations can be done in *Maple*, some steps should still be done by hand.

Additional information about [dsolve](#) is available from the on-line help (?dsolve). Note, in particular, that *dsolve* may return an implicit solution or a solution in parametric form. Explicit solutions can, sometimes, be coerced by using the optional argument *explicit*. Other optional arguments are discussed later in this supplement.

> ?dsolve;
>

▼ Systems of ODEs (Sections 5.2, 5.4 and 5.5 of Nagle/Saff/Snider)

Systems of ODEs are analyzed in a completely parallel manner. The system of equations is specified as a set (with or without initial conditions); the dependent variables are also specified as a set. To illustrate, consider the (linearized) model of a pendulum similar to Project D on page 238:

$\frac{d^2\theta(t)}{dt^2} + 3 \cdot \theta(t) = 0$. This second-order equation is equivalent to the following system of first-order equations:

$$\begin{aligned} &> \text{restart}; \\ &> \text{SYS} := \{ \text{diff}(\theta(t), t) = v(t), \text{diff}(v(t), t) = -3 \cdot \theta(t) \}; \\ &\quad \text{SYS} := \left\{ \frac{d}{dt} \theta(t) = v(t), \frac{d}{dt} v(t) = -3 \theta(t) \right\} \end{aligned} \quad (2.4.1)$$

The dependent variables are:

$$\begin{aligned} &> \text{FNS} := \{ \theta(t), v(t) \}; \\ &\quad \text{FNS} := \{ \theta(t), v(t) \} \end{aligned} \quad (2.4.2)$$

The general solution is found to be

$$\begin{aligned} &> \text{SOL} := \text{dsolve}(\text{SYS}, \text{FNS}); \\ \text{SOL} := \{ \theta(t) = _C1 \sin(\sqrt{3} t) + _C2 \cos(\sqrt{3} t), v(t) = -\sqrt{3} (-\cos(\sqrt{3} t) _C1 \\ &\quad + _C2 \sin(\sqrt{3} t)) \} \end{aligned} \quad (2.4.3)$$

We are now in a position to understand why it is important that **dsolve** returns (a set of) equations. Because a set is unordered, there is no reason to expect the elements of the solution set to appear in the same order each time *Maple* is executed. That is, there is no "first" term in the solution vector. There is, nonetheless, a very simple means to ensure that the appropriate term from the solution is extracted. It is still possible to obtain the individual component as either an expression

$$\begin{aligned} &> \text{SOLV} := \text{subs}(\text{SOL}, v(t)); \\ &\quad \text{SOLV} := -\sqrt{3} (-\cos(\sqrt{3} t) _C1 + _C2 \sin(\sqrt{3} t)) \end{aligned} \quad (2.4.4)$$

or as a function

$$\begin{aligned} &> \text{SOLT} := \text{unapply}(\text{subs}(\text{SOL}, \theta(t)), t); \\ &\quad \text{SOLT} := t \rightarrow _C1 \sin(\sqrt{3} t) + _C2 \cos(\sqrt{3} t) \end{aligned} \quad (2.4.5)$$

Most systems of ODEs do not have explicit solutions. Maple has no difficulty producing series solutions for systems, even nonlinear ones. We illustrate using the corresponding nonlinear pendulum equation.

$$\begin{aligned} &> \text{SYSnl} := \{ \text{diff}(\theta(t), t) = v(t), \text{diff}(v(t), t) = -3 \cdot \sin(\theta(t)) \}; \\ &\quad \text{SYSnl} := \left\{ \frac{d}{dt} \theta(t) = v(t), \frac{d}{dt} v(t) = -3 \sin(\theta(t)) \right\} \end{aligned} \quad (2.4.6)$$

Only a small number of terms will be needed to compare with the linearized solution.

$$> \text{Order} := 3;$$

$$\begin{aligned}
 &> \text{SOLnl} := \text{dsolve}(\text{SYSnl}, \text{FNS}, \text{type} = \text{series}); \\
 \text{SOLnl} &:= \left\{ \theta(t) = \theta(0) + v(0)t - \frac{3}{2} \sin(\theta(0))t^2 + O(t^3), v(t) = v(0) \right. \\
 &\quad \left. - 3 \sin(\theta(0))t - \frac{3}{2} \cos(\theta(0))v(0)t^2 + O(t^3) \right\}
 \end{aligned} \tag{2.4.7}$$

$$\begin{aligned}
 &> \text{SOLnIV} := \text{subs}(\text{SOLnl}, v(t)); \\
 \text{SOLnIV} &:= v(0) - 3 \sin(\theta(0))t - \frac{3}{2} \cos(\theta(0))v(0)t^2 + O(t^3)
 \end{aligned} \tag{2.4.8}$$

Comparisons with the solution to the linearized equation can be performed using the series command, as follows

$$\begin{aligned}
 &> \text{series}(\text{SOLV}, t = 0); \\
 &\quad _C1\sqrt{3} - 3_C2t - \frac{3}{2}_C1\sqrt{3}t^2 + O(t^3)
 \end{aligned} \tag{2.4.9}$$

The real power of Maple for systems of ODEs is seen in its handling of numerical solutions. This procedure will be demonstrated in the next section.

>

▼ Numeric Solutions (Sections 3.6 and 5.3 of Nagle/Saff/Snider)

Numeric solutions to initial value problems can be used in a variety of ways in an introductory course, including direction fields, qualitative behavior of solutions, existence theory, and regularity theory. The **dsolve** command can be used for each of these topics. The default method used is [rkf45](#), a Maple implementation of the Fehlberg fourth-fifth order Runge-Kutta method.

Maple has built-in procedures for solving differential equations numerically. The **dsolve** command with the options **numeric** and **method=classical[choice]** finds a numerical solution by using one of the classical numerical methods. The choice [foreuler](#) implements the classical forward Euler method, the choice [heunform](#) implements the improved Euler's method and the choice [impoly](#) implements the modified Euler method. These methods will be demonstrated in a later section.

To begin our discussion, consider the initial-value problem

$$\begin{aligned}
 &> \text{restart}; \\
 &> \text{ODE} := \text{diff}(x(t), t) = x(t)^2 + t; \text{IC} := x(0) = 0; \text{IVP} := \{ \text{ODE}, \text{IC} \}; \\
 &\quad \text{IVP} := \left\{ \frac{d}{dt} x(t) = x(t)^2 + t, x(0) = 0 \right\}
 \end{aligned} \tag{2.5.1}$$

While this problem does have an explicit solution, it cannot be found using methods typically learned in an introductory course. But, it is reasonable to ask if a solution does exist for all $t > 0$. Numerical solutions can be used, with caution, to discuss questions of this type. The inclusion of the optional argument **type=numeric** is the basic interface to the numerical version of [dsolve](#).

$$\begin{aligned}
 &> \text{numSOLN} := \text{dsolve}(\text{IVP}, x(t), \text{type} = \text{numeric}); \\
 &\quad \text{numSOLN} := \text{proc}(x_rkf45) \dots \text{end proc}
 \end{aligned} \tag{2.5.2}$$

The output from ***dsolve*** is now quite different. What this is saying is that *numSOLN* is a Maple procedure that can be used to compute the solution at a given (numeric) value of t . For example:

```
> numSOLN ( 0 );
```

$$[t = 0., x(t) = 0.] \quad (2.5.3)$$


```
> numSOLN ( 1 / 10 );
```

$$[t = 0.1000000000000000, x(t) = 0.00500052836904142] \quad (2.5.4)$$

Notice that the output from the procedure created by ***dsolve*** is a list of equations with one equation for the independent variable and one equation for each dependent variable. The structure of this solution is ideal for use with the ***subs*** command to extract specific components of a solution or even to use the results in subsequent calculations.

The plot of a numerical solution can be created in a number of different ways. One of the simplest ways is to use the ***odeplot*** command from the ***plots*** package, which must be loaded prior to use. The basic syntax for ***odeplot*** expects the first argument to be the output from numeric ***dsolve***. The second argument should be a list of two or three expressions involving the dependent and independent variables. For example:

```
> with ( plots ) :
> numSOLN := dsolve ( IVP, x(t), type = numeric ) : odeplot ( numSOLN, [t, x(t)], 0..1,
    title = `Approx. solution to x' = x^2 + 1, x(0)=0` );
```




Both numeric *dsolve* and *odeplot* work equally well with higher-order equations and with systems. The basic operations for a higher-order problem will be illustrated using the damped nonlinear pendulum.

```
> ODE := diff( theta(t), t$2 ) + 0.2 * diff( theta(t), t ) + sin(theta(t)) = 0 : IC
    := theta(0) = Pi/4, D(theta)(0) = 0 : IVP := { ODE, IC };
    IVP := {  $\frac{d^2}{dt^2} \theta(t) + 0.2 \left( \frac{d}{dt} \theta(t) \right) + \sin(\theta(t)) = 0, \theta(0) = \frac{1}{4} \pi, D(\theta)(0) = 0 \}$  (2.5.5)
> PEND := dsolve( IVP, theta(t), type = numeric ) :
```


Multiple graphs are included in one plot when the second argument to *odeplot* is a list of lists. (Note how the derivative of the solution is selected using [diff](#).)

```
> odeplot( PEND, [ [ t, theta(t) ], [ t, diff(theta(t), t) ] ], 0..10, title
    = 'Damped Nonlinear Pendulum' );
```

Damped Nonlinear Pendulum


The phase portrait is also easy to create.

```
> odeplot( PEND, [ diff(theta(t), t), theta(t) ], -5..5, title
    = 'Damped Nonlinear Pendulum: phase plane' );
```

Damped Nonlinear Pendulum: phase plane

A three-dimensional view can sometimes be quite illustrative.


```
> odeplot( PEND, [ t, theta(t), diff(theta(t), t) ], 0..25, title
    = 'Damped Nonlinear Pendulum: 3d view', axes = BOXED, color = red );
```

Damped Nonlinear Pendulum: 3d view

You can now click on the image and rotate it by dragging the mouse.

There are times when the standard output from numeric dsolve is not suitable for further processing. The **output = optional** argument provides some flexibility. Specifying **output = listprocedure** will cause **dsolve** to return a *Maple* list of procedures.

```
> PEND2 := dsolve( IVP, theta(t), type = numeric, output = listprocedure );
PEND2 := [ t = proc(t) ... end proc, theta(t) = proc(t) ... end proc,  $\frac{d}{dt} \theta(t) = \text{proc}(t)$ 
...
end proc ]
> THETA := subs( PEND2, theta(t) ); dTHETA := subs( PEND2, diff( theta(t), t ) );
      THETA := proc(t) ... end proc
      dTHETA := proc(t) ... end proc
(2.5.7)
> plot( { THETA, dTHETA }, 0..10, title = 'Damped Nonlinear Pendulum: another view' );
```

Damped Nonlinear Pendulum: another view

The ***output = argument*** is used when the value of the solution is needed only for specified values of the independent variable. This method can be particularly useful when creating a table of data.

```
> PEND3 := dsolve( IVP, theta(t), type = numeric, output = array( [ i/100. $ i = 320 ..330 ] ) );
```

(2.5.8)

$$\begin{aligned}
 PEND3 := & \left[\begin{array}{c} \left[t \quad \theta(t) \quad \frac{d}{dt} \theta(t) \right] \\ \left[\begin{array}{ccc} 3.20000000000000 & -0.562040594090033 & -0.0299450123596072 \\ 3.21000000000000 & -0.562313114439529 & -0.0245601766895983 \\ 3.22000000000000 & -0.562531829642660 & -0.0191840246328481 \\ 3.23000000000000 & -0.562696828567860 & -0.0138169935166268 \\ 3.24000000000000 & -0.562808204453003 & -0.00845951880330290 \\ 3.25000000000000 & -0.562866054905401 & -0.00311203409034493 \\ 3.26000000000000 & -0.562870481901806 & 0.00222502888967990 \\ 3.27000000000000 & -0.562821591788409 & 0.00755124026910515 \\ 3.28000000000000 & -0.562719495280840 & 0.0128661720451644 \\ 3.29000000000000 & -0.562564307464168 & 0.0181693980799930 \\ 3.30000000000000 & -0.562356147436138 & 0.0234604938145919 \end{array} \right] \end{array} \right] \quad (2.5.8)
 \end{aligned}$$

Extracting information from this table can be frustrating. Note that the structure of *PEND3* is a two-dimensional (column) array. The first element of *PEND3* is the three-dimensional (row) vector which describes the contents of the matrix contained in the second element of *PEND3*.

> *VAR3* := evalm(*PEND3*[1, 1]);

$$\text{VAR3} := \left[t \quad \theta(t) \quad \frac{d}{dt} \theta(t) \right] \quad (2.5.9)$$

> *VAR3*[2];

$$\theta(t) \quad (2.5.10)$$

It is also possible to nest the indices to directly access entries in this data structure. For example, the pendulum is at rest sometime between $t = 3.25$ and $t = 3.26$. At this time, the angle describing the pendulum's position is approximately

$$\begin{aligned}
 > \text{rest} := (\text{PEND3}[2, 1][6, 2] + \text{PEND3}[2, 1][7, 2]) / 2; \\
 &\quad \text{rest} := -0.562868268403604 \quad (2.5.11)
 \end{aligned}$$

which, converted to degrees, is

$$\begin{aligned}
 > \text{evalf}(\text{rest} \cdot 180 / \text{Pi}); \\
 &\quad -32.2499761928740 \quad (2.5.12)
 \end{aligned}$$


To illustrate the use of numeric ***dsolve*** for a system, consider the second-order system

$$\begin{aligned}
 > \text{ODE} := \text{diff}(x(t), t) = 3 \cdot x(t)^2 \cdot y(t) - 6 \cdot y(t)^2, \text{diff}(y(t), t) = -x(t)^3 + 2 \\
 &\quad \cdot x(t)^2 + 2 \cdot x(t) \cdot y(t) - 4 \cdot y(t) : \text{IC} := x(0) = 1, y(0) = 0 : \text{IVP} := \{ \text{ODE}, \\
 &\quad \text{IC} \}; \\
 \text{IVP} := & \left\{ \frac{d}{dt} x(t) = 3 x(t)^2 y(t) - 6 y(t)^2, \frac{d}{dt} y(t) = -x(t)^3 + 2 x(t)^2 \right. \quad (2.5.13)
 \end{aligned}$$

$$+ 2x(t)y(t) - 4y(t), x(0) = 1, y(0) = 0 \}$$

The level curves of the (Liapunov) function $L(x, y) = (x - 2)^2 + 3y^2$ are trajectories for this system. That is, $L(x(t), y(t)) = L(x_0, y_0)$ for all $t > 0$. This can be illustrated using numeric [dsolve](#) and [odeplot](#) as follows.

```
> numSOLN := dsolve( IVP, [ x(t), y(t) ], type = numeric ) : odeplot( numSOLN, [ t,
    (x(t)-2)^2 + 3*y(t)^2 ], 0..10, title = `Values of L(x,y) = (x-2)^2 + 3y^2`, view
    = [0..10, 0..2] );
```



```
> ?dsolve
> ?dsolve[numeric]
> ?dsolve[dverk78]
> ?plots[odeplot]
>
```


▼ Series Solutions (*Chapter 8 of Nagle/Saff/Snider*)

Some differential equations, and systems, are most approachable using power series techniques. *Maple* provides access to this method by way of the **type = series** optional argument to [dsolve](#).

Consider the differential equation, in normal form,

$$\begin{aligned} &> \text{ODE} := \text{diff}(y(x), x^2) + 1/x \cdot \text{diff}(y(x), x) - 1/(x^2 + 1) \cdot y(x) = 0; \\ &\text{ODE} := \frac{d^2}{dx^2} y(x) + \frac{\frac{d}{dx} y(x)}{x} - \frac{y(x)}{x^2 + 1} = 0 \end{aligned} \quad (2.6.1)$$

The first few terms of the power series, centered at $x = 0$, can be obtained with the command

$$\begin{aligned} &> \text{dsolve}(\text{ODE}, y(x), \text{type} = \text{series}); \\ y(x) = &_C1 \left(1 + \frac{1}{4} x^2 - \frac{3}{64} x^4 + O(x^6) \right) + _C2 \left(\ln(x) \left(1 + \frac{1}{4} x^2 - \frac{3}{64} x^4 \right. \right. \\ &\left. \left. + O(x^6) \right) + \left(-\frac{1}{4} x^2 + \frac{1}{128} x^4 + O(x^6) \right) \right) \end{aligned} \quad (2.6.2)$$

To obtain additional terms in the sequence, modify the system constant [Order](#).

$$\begin{aligned} &> \text{Order} := 10 : \\ &> Y9 := \text{dsolve}(\text{ODE}, y(x), \text{type} = \text{series}); \\ Y9 := y(x) = &_C1 \left(1 + \frac{1}{4} x^2 - \frac{3}{64} x^4 + \frac{5}{256} x^6 - \frac{175}{16384} x^8 + O(x^{10}) \right) \\ &+ _C2 \left(\ln(x) \left(1 + \frac{1}{4} x^2 - \frac{3}{64} x^4 + \frac{5}{256} x^6 - \frac{175}{16384} x^8 + O(x^{10}) \right) + \left(-\frac{1}{4} x^2 + \frac{1}{128} x^4 + \frac{1}{1536} x^6 - \frac{265}{196608} x^8 + O(x^{10}) \right) \right) \end{aligned} \quad (2.6.3)$$

Note that $x = 0$ is a regular singular point for this equation. Note also the structure of this solution. The presence of the logarithmic factor is enlightening; that same term also complicates the conversion of this solution from an equation whose RHS is of type series into an expression that can be plotted, etc.

Series solutions centered at other points can be requested by specifying a full set of initial conditions at the desired value of the independent variable. Here, the series expansion of the solution centered at $x = -1$ will be produced.

$$\begin{aligned} &> \text{IVP} := \{ \text{ODE}, y(-1) = c_1, D(y)(-1) = c_2 \}; \\ \text{IVP} := &\left\{ \frac{d^2}{dx^2} y(x) + \frac{\frac{d}{dx} y(x)}{x} - \frac{y(x)}{x^2 + 1} = 0, y(-1) = c_1, D(y)(-1) = c_2 \right\} \end{aligned} \quad (2.6.4)$$

> Order := 6 :

> Y5 := dsolve(IVP, y(x), type = series);

(2.6.5)

$$\begin{aligned}
 Y5 := y(x) = & c_1 + c_2 (1+x) + \left(\frac{1}{2} c_2 + \frac{1}{4} c_1 \right) (1+x)^2 + \left(\frac{5}{12} c_2 + \frac{1}{6} c_1 \right) (1+x)^3 \\
 & + \left(\frac{1}{3} c_2 + \frac{11}{96} c_1 \right) (1+x)^4 + \left(\frac{127}{480} c_2 + \frac{1}{12} c_1 \right) (1+x)^5 \\
 & + O((1+x)^6)
 \end{aligned} \tag{2.6.5}$$

Here are two ways of obtaining the series solutions for two linearly independent solutions to this ODE. First, it is possible to simply specify two different sets of values for the constants c_1 and c_2 .

> $Y5a := \text{subs}(c_1 = 1, c_2 = 0, \text{rhs}(Y5)); Y5b := \text{subs}(c_1 = 0, c_2 = 1, \text{rhs}(Y5));$

$$\begin{aligned}
 Y5a := & 1 + \frac{1}{4} (1+x)^2 + \frac{1}{6} (1+x)^3 + \frac{11}{96} (1+x)^4 + \frac{1}{12} (1+x)^5 + O((1+x)^6) \\
 Y5b := & 1 + x + \frac{1}{2} (1+x)^2 + \frac{5}{12} (1+x)^3 + \frac{1}{3} (1+x)^4 + \frac{127}{480} (1+x)^5 \\
 & + O((1+x)^6)
 \end{aligned} \tag{2.6.6}$$

Alternatively, *Maple* can separately collect the terms involving c_1 and c_2 . But, before this collect can be used, the solution must be converted to a polynomial. (As long as the O term is present, the solution is an equation whose RHS is of type series.)

> $Y5 := \text{convert}(\text{rhs}(Y5), \text{polynom});$


$$\begin{aligned}
 Y5 := & c_1 + c_2 (1+x) + \left(\frac{1}{2} c_2 + \frac{1}{4} c_1 \right) (1+x)^2 + \left(\frac{5}{12} c_2 + \frac{1}{6} c_1 \right) (1+x)^3 \\
 & + \left(\frac{1}{3} c_2 + \frac{11}{96} c_1 \right) (1+x)^4 + \left(\frac{127}{480} c_2 + \frac{1}{12} c_1 \right) (1+x)^5
 \end{aligned} \tag{2.6.7}$$

> $Y5 := \text{collect}(Y5, [c_1, c_2]);$

$$\begin{aligned}
 Y5 := & \left(\frac{1}{4} (1+x)^2 + \frac{1}{6} (1+x)^3 + \frac{11}{96} (1+x)^4 + 1 + \frac{1}{12} (1+x)^5 \right) c_1 \\
 & + \left(\frac{5}{12} (1+x)^3 + 1 + x + \frac{1}{2} (1+x)^2 + \frac{127}{480} (1+x)^5 + \frac{1}{3} (1+x)^4 \right) c_2
 \end{aligned} \tag{2.6.8}$$

Note that the ordering of the terms in this last expression is less than optimal. (Controlling the order of terms in a polynomial is not a simple task in *Maple*.)

> $\text{plot}(\text{map}(\text{convert}, \{Y5a, Y5b\}, \text{polynom}), x = -2..1, \text{title} = \text{'Linearly Independent Solution'});$

Linearly Independent Solution

>

▼ **Laplace Transforms (Chapter 7 of Nagle/Saff/Snider)**

All uses of Laplace transforms should begin by reading the integral transform library into the Maple session.

```
> with(inttrans);
[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, (2.7.1)
 invmellin, laplace, mellin, savetable]
```

The two commands that form the Laplace transform pair are [laplace](#) and [invlaplace](#). These commands can be used explicitly or implicitly to solve a differential equation. To illustrate both approaches, consider the problem of determining the general solution to

```
> ODE := diff( y(t), t$2 ) + 3·diff( y(t), t ) - 4·y(t) = 0;
```

$$ODE := \frac{d^2}{dt^2} y(t) + 3 \left(\frac{d}{dt} y(t) \right) - 4 y(t) = 0 \quad (2.7.2)$$

One approach is to begin by applying the Laplace transform to this equation and solving for the Laplace transform, $Y(s)$, of the solution $y(t)$.

$$\begin{aligned} &> \text{lapODE} := \text{laplace}(\text{ODE}, t, s); \\ \text{lapODE} &:= s^2 \text{laplace}(y(t), t, s) - D(y)(0) - s y(0) + 3 s \text{laplace}(y(t), t, s) \\ &\quad - 3 y(0) - 4 \text{laplace}(y(t), t, s) = 0 \end{aligned} \quad (2.7.3)$$

$$\begin{aligned} &> \text{SOLY} := \text{solve}(\text{lapODE}, \text{laplace}(y(t), t, s)); \\ \text{SOLY} &:= \frac{3 y(0) + D(y)(0) + s y(0)}{s^2 + 3 s - 4} \end{aligned} \quad (2.7.4)$$

Next, the inverse Laplace transform is applied to find the solution $y(t)$.

$$\begin{aligned} &> \text{SOLy} := \text{invlaplace}(\text{SOLY}, s, t); \\ \text{SOLy} &:= \frac{1}{5} e^{-\frac{3}{2}t} \left(5 y(0) \cosh\left(\frac{5}{2}t\right) + \sinh\left(\frac{5}{2}t\right) (2 D(y)(0) + 3 y(0)) \right) \end{aligned} \quad (2.7.5)$$

or, as follows, in a form that is sometimes more useful.

$$\begin{aligned} &> \text{collect}(\text{SOLy}, [y(0), D(y)(0)]); \\ \frac{1}{5} e^{-\frac{3}{2}t} &\left(5 \cosh\left(\frac{5}{2}t\right) + 3 \sinh\left(\frac{5}{2}t\right) \right) y(0) + \frac{2}{5} e^{-\frac{3}{2}t} \sinh\left(\frac{5}{2}t\right) D(y)(0) \end{aligned} \quad (2.7.6)$$

Alternatively, a one-step solution is possible with the use of the **method = laplace** optional argument to [dsolve](#).

$$\begin{aligned} &> \text{dsolve}(\text{ODE}, y(t), \text{method} = \text{laplace}); \\ y(t) &= \frac{1}{5} e^{-\frac{3}{2}t} \left(5 y(0) \cosh\left(\frac{5}{2}t\right) + \sinh\left(\frac{5}{2}t\right) (2 D(y)(0) + 3 y(0)) \right) \end{aligned} \quad (2.7.7)$$

For additional assistance, consult the on-line help for [dsolve](#)

as well as the help pages for the individual commands

```
> ?intrans[laplace];
> ?intrans[invlaplace];
>
```